NAG Library Routine Document

F07CAF (DGTSV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07CAF (DGTSV) computes the solution to a real system of linear equations

$$AX = B,$$

where A is an n by n tridiagonal matrix and X and B are n by r matrices.

2 Specification

```
SUBROUTINE F07CAF (N, NRHS, DL, D, DU, B, LDB, INFO)
INTEGER                 N, NRHS, LDB, INFO
REAL (KIND=nag_wp) DL(*), D(*), DU(*), B(LDB,*)
```

The routine may be called by its LAPACK name ***dgtsv***.

3 Description

F07CAF (DGTSV) uses Gaussian elimination with partial pivoting and row interchanges to solve the equations $AX = B$. The matrix A is factorized as $A = PLU$, where P is a permutation matrix, L is unit lower triangular with at most one nonzero subdiagonal element per column, and U is an upper triangular band matrix, with two superdiagonals.

Note that the equations $A^{\mathrm{T}}X = B$ may be solved by interchanging the order of the arguments DU and DL.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of linear equations, i.e., the order of the matrix A.

Constraint: N ≥ 0.

2: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrix B.

Constraint: NRHS ≥ 0.

3: DL($*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array DL must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain the $(n - 1)$ subdiagonal elements of the matrix A.

On exit: if no constrains are violated, DL is overwritten by the $(n-2)$ elements of the second superdiagonal of the upper triangular matrix U from the LU factorization of A, in $\mathrm{DL}(1), \mathrm{DL}(2), \ldots, \mathrm{DL}(n-2)$.

4: D$(*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the matrix A.

On exit: if no constraints are violated, D is overwritten by the n diagonal elements of the upper triangular matrix U from the LU factorization of A.

5: DU$(*)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the dimension of the array DU must be at least $\max(1, \mathrm{N}-1)$.

On entry: must contain the $(n-1)$ superdiagonal elements of the matrix A.

On exit: if no constraints are violated, DU is overwritten by the $(n-1)$ elements of the first superdiagonal of U.

6: B$(\mathrm{LDB}, *)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

Note: To solve the equations $Ax = b$, where b is a single right-hand side, B may be supplied as a one-dimensional array with length $\mathrm{LDB} = \max(1, \mathrm{N})$.

On entry: the n by r right-hand side matrix B.

On exit: if $\mathrm{INFO} = 0$, the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F07CAF (DGTSV) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

8: INFO – INTEGER *Output*

On exit: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$\mathrm{INFO} < 0$

If $\mathrm{INFO} = -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

$\mathrm{INFO} > 0$

Element $\langle value \rangle$ of the diagonal is exactly zero, and the solution has not been computed. The factorization has not been completed unless $\mathrm{N} = \langle value \rangle$.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the ***machine precision***. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x}-x\|_1}{\|x\|_1} \leq \kappa(A)\frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \left\|A^{-1}\right\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

Alternatives to F07CAF (DGTSV), which return condition and error estimates are F04BCF and F07CBF (DGTSVX).

8 Parallelism and Performance

Not applicable.

9 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ is proportional to nr.

The complex analogue of this routine is F07CNF (ZGTSV).

10 Example

This example solves the equations

$$Ax = b,$$

where A is the tridiagonal matrix

$$A = \begin{pmatrix} 3.0 & 2.1 & 0 & 0 & 0 \\ 3.4 & 2.3 & -1.0 & 0 & 0 \\ 0 & 3.6 & -5.0 & 1.9 & 0 \\ 0 & 0 & 7.0 & -0.9 & 8.0 \\ 0 & 0 & 0 & -6.0 & 7.1 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 2.7 \\ -0.5 \\ 2.6 \\ 0.6 \\ 2.7 \end{pmatrix}.$$

10.1 Program Text

```
    Program f07cafe

!     F07CAF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: dgtsv, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: info, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: b(:), d(:), dl(:), du(:)
!     .. Executable Statements ..
      Write (nout,*) 'F07CAF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n

      Allocate (b(n),d(n),dl(n-1),du(n-1))

!     Read the tridiagonal matrix A and the right hand side B from
!     data file
```

```
      Read (nin,*) du(1:n-1)
      Read (nin,*) d(1:n)
      Read (nin,*) dl(1:n-1)
      Read (nin,*) b(1:n)

!     Solve the equations Ax = b for x

!     The NAG name equivalent of dgtsv is f07caf
      Call dgtsv(n,1,dl,d,du,b,n,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Solution'
        Write (nout,99999) b(1:n)

      Else
        Write (nout,99998) 'The (', info, ',', info, ')', &
          ' element of the factor U is zero'
      End If

99999 Format ((1X,7F11.4))
99998 Format (1X,A,I3,A,I3,A,A)
    End Program f07cafe
```

10.2 Program Data

```
F07CAF Example Program Data
  5                              :Value of N
        2.1  -1.0   1.9   8.0
  3.0   2.3  -5.0  -0.9   7.1
  3.4   3.6   7.0  -6.0         :End of matrix A
  2.7  -0.5   2.6   0.6   2.7  :End of vector B
```

10.3 Program Results

```
 F07CAF Example Program Results

 Solution
     -4.0000     7.0000     3.0000    -4.0000    -3.0000
```